<commit_before>type
  BaseObj = object of RootObj
  DerivedObj = object of BaseObj

  Container[T] = object

proc doSomething(c: Container[BaseObj or DerivedObj]) = discard

var t: Container[DerivedObj]
doSomething t
<commit_msg>more advanced fake covariance<commit_after>template accept(x) =
  static: assert(compiles(x))

template reject(x) =
  static: assert(not compiles(x))

type
  BaseObj = object of RootObj
  DerivedObj = object of BaseObj
  NonDerivedObj = object

  Container[T] = object

var base: BaseObj
var derived: DerivedObj
var nonDerived: NonDerivedObj

var baseContainer: Container[BaseObj]
var derivedContainer: Container[DerivedObj]
var nonDerivedContainer: Container[NonDerivedObj]

# We can fake covariance by listing some specific derived types that
# will be allowed with our overload. This is not a real covariance,
# because there will be multiple instantiations of the proc, but for
# many purposes, it will suffice:

proc wantsSpecificContainers(c: Container[BaseObj or DerivedObj]) = discard

accept wantsSpecificContainers(baseContainer)
accept wantsSpecificContainers(derivedContainer)

reject wantsSpecificContainers(nonDerivedContainer)
reject wantsSpecificContainers(derived)

# Now, let's make a more general solution able to catch all derived types:

type
  DerivedFrom[T] = concept type A
    var derived: ref A
    var base: ref T = derived

proc wantsDerived(x: DerivedFrom[BaseObj]) = discard

accept wantsDerived(base)
accept wantsDerived(derived)

reject wantsDerived(nonDerived)
reject wantsDerived(baseContainer)

proc wantsDerivedContainer(c: Container[DerivedFrom[BaseObj]]) = discard

accept wantsDerivedContainer(baseContainer)
accept wantsDerivedContainer(derivedContainer)

reject wantsDerivedContainer(nonDerivedContainer)

{'fraction_non_alphanumeric': 0.0647887323943662, 'fraction_numerical': 0.0, 'mean_word_length': 6.493670886075949, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2958033', 'n_tokens_mistral': 522, 'n_tokens_neox': 474, 'n_words': 167}